International Mathematical Forum, 4, 2009, no. 45, 2217 - 2222

God's Algorithm for Rubik's Cube
An Integer Programming Approach

Cihan Aksop

Gazi University, Department of Statistics
06500 Teknikokullar, Ankara, Turkey
entelpi@yahoo.com

Abstract

An integer mathematical model is proposed to find the best strategy (God's algorithm) to solve a randomly scrambled Rubik's cube is presented.

Mathematics Subject Classification: 90C10

Keywords: God's algorithm, Rubik's cube, integer programming

1 Introduction

Rubik's cube is invented in 1970's by Erno Rubik. This cube contains (in standart version) 3 x 3 x 3 different colored subcubes. As mentioned in [2] at each move any of 3 x 3 x 1 plane of cube can be rotated 90, 180 or 270 degrees relatively to the rest of cube. The aim of the game is to find the sequences of moves that yields each side of cube are the same color. The problem studied in this paper is to find a solution strategy -sequences of moves- which requires fewest number of moves to reach that aim. This strategy is called "God's algorithm". There are several algorithms which are proposed by Rokicki [4], Korf [2], Kociemba [1] and many others but non of them has been proved as God's Algorithm.

2 Mathematical Models

A standart 3 x 3 x 3 Rubik's cube can be figured as Figure 1. We labeled all the subcubes 1 to 54 and we will call that *map of Rubik's cube*. In Figure 1, A - I and I - XII represents the line labels. We will use +, - upper-scribes to line labels for responding move: for example A^+ means a left-to-right 90

- + ↔	I	II	III	IV	V	VI	VII	VIII	IX	X	XI	XII
A				1	2	3						
B				4	5	6						
C				7	8	9						
D	10	11	12	13	14	15	16	17	18	19	20	21
E	22	23	24	25	26	27	28	29	30	31	32	33
F	34	35	36	37	38	39	40	41	42	43	44	45
G				46	47	48						
H				49	50	51						
I				52	53	54						

Figure 1: The map of Rubik's Cube

degree turn of line A, XII^- means an up-to-down 90 degree turn of line XII. Observe that more then one line can be effected by making a single move; for example, E^+ effects only E line but V^- effects both line V and XI.

We use two set of decision variables. First set of decision variables controls the current patern of the cube and is defeined as follows:

$x_{i,t}$: The color index of subcube at the map of Rubik's cube labeled by i at t. turn

Since there are six different colors in a Rubik's cube, $x_{i,t}$ can take integer values 1 to 6.

The second set of decision variables controls the moves and is defined as follows:

$$y_{i,t} = \begin{cases} 1 & ,\text{if } t \text{ turn is an } i \text{ move} \\ 0 & ,\text{otherwise} \end{cases} \quad (1)$$

where ith move defiened as follows:

i	1	2	3	4	5	6	7	8	9
move	A^+	A^-	B^+	B^-	C^+	C^-	D^+	D^-	E^+
i	10	11	12	13	14	15	16	17	18
move	E^-	F^+	F^-	IV^+	IV^-	V^+	V^-	VI^+	VI^-

Figure 2: Description of ith move

So $y_{3,t}$ means tth move is B^+.

Let us define a new set called *Gecis* as follows:

$$Gecis = \{(k,i,j) \in \{1,\ldots,18\} \times \{1,\ldots,54\}^2 : \text{subcube at position } i \text{ in the map of Rubik's Cube will go to the position } j \text{ if a } k \text{ move done}\}$$

Then the mathematical model for God's algorithm is as follows:

$$\min \sum_{i=1}^{18} \sum_{t=1}^{54} t y_{i,t} \tag{2}$$

$$\forall t = 1, \ldots, 53, \quad \forall (k, i, j) \in Gecis$$

$$x_{i,t} - 6(1 - y_{k,t}) \leq x_{j,t+1} \leq x_{i,t} + 6(1 - y_{k,t}) \tag{3}$$

$$x_{j,t} - 6(1 - y_{k+1,t}) \leq x_{i,t+1} \leq x_{j,t} + 6(1 - y_{k+1,t}) \tag{4}$$

$$x_{i,t} - 6\left[y_{k,t} + y_{k+1,t} + \sum_{\substack{(l,i,n)\in Gecis \\ l\neq k}} (y_{l,t} + y_{l+1,r})\right] \leq x_{i,t+1}$$
$$\leq x_{i,t} + 6\left[y_{k,t} + y_{k+1,t} + \sum_{\substack{(l,i,n)\in Gecis \\ l\neq k}} (y_{l,t} + y_{l+1,t})\right] \tag{5}$$

$$\forall t = 1, \ldots, 54$$

$$\sum_{i=1}^{18} y_{i,t} \leq 1 \tag{6}$$

A new pattern on the map of Rubik's Cube is achieved with the constaints (3) and (4) if a positive or negative move applied to $x_{i,t}$ repectively. Constraints (5) makes $x_{i,t} = x_{i,t+1}$ if the i. subcube do not move at tth turn. Last constraints satisfied that one turn per move.

A glpk Code for God's Algorithm

```
set gecis, dimen 3;
set ayniust;
set aynisol;
set aynion;
set aynisag;
set ayniarka;
set aynialt;
param baslangicdeseni{i in 1..54}, integer, >0;
```

```
var x{1..54,1..54},integer,<=6,>=0;
var y{1..18,1..54},integer,<=1,>=0;
minimize haraket:sum{i in 1..18,t in 1..54} t*y[i,t];
#initial pattern
s.t. baslangic1{i in 1..54}:x[i,1]=baslangicdeseni[i];
#...
# + move
s.t. kisitbir{t in 1..53,(k,i,j) in gecis}:
x[i,t]-6*(1-y[k,t])<=x[j,t+1];
s.t. kisitiki{t in 1..53,(k,i,j) in gecis}:
x[j,t+1]<=x[i,t]+6*(1-y[k,t]);

# - move
s.t. kisituc{t in 1..53,(k,j,i) in gecis}:
x[i,t]-6*(1-y[k+1,t])<=x[j,t+1];
s.t. kisitdort{t in 1..53,(k,j,i) in gecis}:
x[j,t+1]<=x[i,t]+6*(1-y[k+1,t]);

# subcubes without move
s.t. kisitbes{t in 1..53,(k,i,j) in gecis}:
x[i,t]-6*(y[k,t]+y[k+1,t]+sum{(l,m,n) in gecis:m=i and k<>l}
 (y[l,t] + y[l+1,t]))<=x[i,t+1];
s.t. kisitalti{t in 1..53,(k,i,j) in gecis}:
x[i,t+1]<=x[i,t]+6*(y[k,t]+y[k+1,t]
+sum{(l,m,n) in gecis:m=i and k<>l}(y[l,t]+y[l+1,t]));

# one move per turn
s.t. kisityedi{t in 1..54}:sum{i in 1..18} y[i,t]<= 1;

# subcubes' color must be same for each side
s.t. kisitust{i in ayniust,j in ayniust:i>j}:x[i,54]=x[j,54];
s.t. kisitsol{i in aynisol,j in aynisol:i>j}:x[i,54]=x[j,54];
s.t. kisiton{i in aynion,j in aynion:i>j}:x[i,54]=x[j,54];
s.t. kisitsag{i in aynisag,j in aynisag:i>j}:x[i,54]=x[j,54];
s.t. kisitarka{i in ayniarka,j in ayniarka:i>j}:x[i,54]=x[j,54];
s.t. kisitalt{i in aynialt,j in aynialt:i>j}:x[i,54]=x[j,54];

data;
set gecis := (1,1,18) (1,2,30) (1,3,42) (1,10,3) (1,22,2)
(1,34,1) (1,18,54) (1,30,53) (1,42,52) (1,52,10) (1,53,22)
(1,54,34) (1,19,43) (1,20,31) (1,21,19) (1,31,44) (1,33,20)
(1,43,45) (1,44,33) (1,45,21) (3,4,17) (3,5,29) (3,6,41)
```

```
(3,11,6) (3,23,5) (3,35,4) (3,17,51) (3,29,50) (3,41,49)
(3,49,11) (3,50,23) (3,51,35) (5,7,16) (5,8,28) (5,9,40)
(5,12,9) (5,24,8) (5,36,7) (5,16,48) (5,28,47) (5,40,46)
(5,46,12) (5,47,24) (5,48,36) (5,13,15) (5,14,27) (5,15,39)
(5,25,14) (5,27,38) (5,37,13) (5,38,25) (5,39,37) (7,10,13)
(7,11,14) (7,12,15) (7,13,16) (7,14,17) (7,15,18) (7,16,19)
(7,17,20) (7,18,21) (7,19,10) (7,20,11) (7,21,12) (7,1,3)
(7,2,6) (7,3,9) (7,4,2) (7,6,8) (7,7,1) (7,8,4) (7,9,7)
(9,22,25) (9,23,26) (9,24,27) (9,25,28) (9,26,29) (9,27,30)
(9,28,31) (9,29,32) (9,30,33) (9,31,22) (9,32,23) (9,33,24)
(11,34,37) (11,35,38) (11,36,39) (11,37,40) (11,38,41)
(11,39,42) (11,40,43) (11,41,44) (11,42,45) (11,43,34)
(11,44,35) (11,45,36) (11,46,48) (11,47,51) (11,48,54)
(11,49,47) (11,51,53) (11,52,46) (11,53,49) (11,54,52)
(13,1,45) (13,4,33) (13,7,21) (13,13,1) (13,25,4) (13,37,7)
(13,21,52) (13,33,49) (13,45,46) (13,46,13) (13,49,25)
(13,52,37) (13,10,34) (13,11,22) (13,12,10) (13,22,35)
(13,24,11) (13,34,36) (13,35,24) (13,36,12) (15,2,44) (15,5,32)
(15,8,20) (15,14,2) (15,26,5) (15,38,8) (15,20,53) (15,32,50)
(15,44,47) (15,47,14) (15,50,26) (15,53,38) (17,3,43) (17,6,31)
(17,9,19) (17,15,3) (17,27,6) (17,39,9) (17,19,54) (17,31,51)
(17,43,48) (17,48,15) (17,51,27) (17,54,39) (17,16,18)
(17,17,30) (17,18,42) (17,28,17) (17,30,41) (17,40,16)
(17,41,28) (17,42,40);
set ayniust := 1,2,3,4,5,6,7,8,9;
set aynisol := 10,11,12,22,23,24,34,35,36;
set aynion := 13,14,15,25,26,27,37,38,39;
set aynisag := 16,17,18, 28,29,30,40,41,42;
set ayniarka := 19,20,21,31,32,33,43,44,45;
set aynialt := 46,47,48,49,50,51,52,53,54;

#baslangicdeseni is the initial pattern to be defiened
param baslangicdeseni := 1 6, 2 6, 3 3, 4 3, 5 6, 6 6, 7 4, 8 5,
9 5, 10 1, 11 2, 12 2, 13 3, 14 3, 15 3, 16 2, 17 1, 18 5, 19 6,
20 3, 21 5, 22 1, 23 3, 24 2, 25 5, 26 5, 27 6, 28 5, 29 1, 30 1,
31 5, 32 4, 33 2, 34 2, 35 4, 36 1, 37 6, 38 4, 39 3, 40 4, 41 1,
42 2, 43 1, 44 2, 45 1, 46 4, 47 3, 48 6, 49 6, 50 2, 51 4, 52 5,
53 4, 54 4;

end;
```

References

[1] Kociemba, H. Close to God's algorithm. *Cubism for Fun* 1992, 10–13.

[2] Korf, R. E. Finding optimal solution to Rubik's Cube using pattern databases. *Proceedings of the Fourteenth National Conference on Artifical Intelligence.*

[3] Kunkle, D., Cooperman, G. Twenty-siz moves suffice for Rubik's Cube. *Proceeding's of International Symposium on Symbolic and Algebraic Computation*, AIM Press, 2007, 235–242.

[4] Rokicki, T. Twnety-five moves suffice for Rubik's Cube. arXiv: 0803.3435v1, http://arxiv.org/abs/0803.3435.

Received: April, 2009